<!--#include file="lib/utf-8.asp"-->
<!--#include file="lib/connopen.asp"-->
<!--#include file="lib/asp-common-lib.asp"-->
<%
tx_search=trim(SafeRequest("tx_search",0))
status=trim(SafeRequest("status",0))
swv=trim(SafeRequest("swv",0))
brv=trim(SafeRequest("brv",0))
bm=trim(SafeRequest("bm",0))
bv=trim(SafeRequest("bv",0))
ksrq=trim(SafeRequest("ksrq",0))
' if ksrq<>"" then ksrq=cdate(ksrq)
jsrq=trim(SafeRequest("jsrq",0))
' if jsrq<>"" then jsrq=cdate(jsrq)
search_count=trim(SafeRequest("search_count",1))
'组合sql查询字符串
'sql="select * from cdfee where true "
sql="select "
sql=sql&" * from bnd where 1 "

if tx_search<>"" then
    sql=sql&" and (ndip like '%"&tx_search&"%' or sEsn like '%"&tx_search&"%' or sMac like '%"&tx_search&"%')"
end if
if ksrq<>"" and jsrq<>"" then sql=sql&" and (date(dtLog)>='"&ksrq&"' and date(dtLog)<='"&jsrq&"')"
if status<>"" then sql=sql&" and sStatus='"&status&"'"
if swv<>"" then sql=sql&" and sSoftwareVer='"&swv&"'"
if brv<>"" then sql=sql&" and sBootromVer='"&brv&"'"
if bv<>"" then sql=sql&" and sBoardVer='"&bv&"'"
if bm<>"" then sql=sql&" and sBoardModel='"&bm&"'"
sql=sql&" order by dtLog desc"
if search_count<>0 then sql=sql&" limit "&search_count

' response.write sql
' response.end
'执行数据查询
set rs = Server.CreateObject("ADODB.Recordset")
rs.cursorlocation=3
rs.open sql,conn,1,1
If rs.eof Then
  response.write("没有检索到数据!")
  response.end
End if

'分页控制
If saferequest("page",0)<>"" Then page=clng(saferequest("page",0)) Else page=1
pc=saferequest("pc",1)
If pc="" Then pc=25
rs.pagesize=pc
pagesize=rs.pagesize
pagecount=rs.pagecount
'page=clng(request.QueryString("page"))
if Not IsEmpty(page) then '如果page_num已经初始化...
  if Not IsNumeric(page) then '判断page_num值是否为数字
    page=1
  else
    page = clng(page) '接收page_num并化为数字型赋给page_num变量
  end if
  if page > pagecount then '如果接收的页数大于总页数
    rs.AbsolutePage = pagecount '设置当前显示页等于最后页
  elseif page <= 0 then '如果page小于等于0
    page=1
    rs.AbsolutePage = page '设置当前显示页等于第一页
  else
    rs.AbsolutePage = page '如果大于零,显示当前页等于接收的页数
  end if
  page = rs.AbsolutePage
end if
%>

<table id="tb_poll" class="full">
  <thead>
    <tr>
      <th></th>
      <th>IP Address</th>
      <th>Log Time</th>
      <th>Reboot</th>
      <th>Status</th>
      <th>SoftVer</th>
      <th>BootVer</th>
      <th>RunTime</th>
      <th>BoardModel</th>
      <th>BoardVer</th>
      <th>Esn</th>
      <th>Mac</th>
      <th>Flash</th>
      <th>Sdram</th>
    </tr>
  </thead>
  <tbody>
  <%
  For i=1 To pagesize
    If rs.eof Then Exit for
  %>
    <tr>
      <td><%=rs.AbsolutePosition%></td>
      <td><a class="res" href="<%=rs("ndip")%>" rel="ndinfo.asp?ndip=<%=rs("ndip")%>"><%=rs("ndip")%></a></td>
      <td><a target="_blank" href="snmpbadinfo.asp?id=<%=rs("id")%>"><%=rs("dtLog")%></a></td>
      <td><%=rs("bReboot")%></td>
      <td><%=rs("sStatus")%></td>
      <td><%=rs("sSoftwareVer")%></td>
      <td><%=rs("sBootromVer")%></td>
      <td><%=rs("sRunningTime")%></td>
      <td><%=rs("sBoardModel")%></td>
      <td><%=rs("sBoardVer")%></td>
      <td><%=rs("sEsn")%></td>
      <td><%=rs("sMac")%></td>
      <td><%=rs("sFlashSize")%></td>
      <td><%=rs("sSdramSize")%></td>
    </tr>
  <%
    rs.movenext
  next
  %>
  </tbody>
</table>
<div>
  <span id="pagebar">
    共查询到&nbsp;<%=rs.RecordCount%>&nbsp;条记录&nbsp;
    <%If page>1 Then%>
      [<a href="#" id="1">首页</a>]
      [<a href="#" id="<%=page-1%>">上一页</a>]
    <%
        End If
        If Not (page>=pagecount Or pagecount=0) Then
    %>
      [<a href="#" id="<%=page+1%>">下一页</a>]
      [<a href="#" id="<%=pagecount%>">尾页</a>]
    <%End If %>
      第 <%=page%> 页 / 共 <%=pagecount%> 页
      转到<input id="page_val" type="text" size="2" maxlength="<%=Int(Log(pagecount))+1%>" value="<%=page%>" />页
      <input id="page_go" class="bt_button" type="button" value="Go" />
  </span>
</div>
<%
rs.close
Set rs=nothing
%>
<!--#include file="lib/connclose.asp"-->

